

Hexaly Optimizer

(Hexaly Optimizer14.5)

使用手引書

MSI 株式会社

2026/3/16

内容

1. Hexaly Optimizer とは.....	2
2. Healy Optimizer の実行方法 (稼働確認)	2
2.1 hxm ファイルの実行	3
2.2 実行結果.....	5
3. Hexaly Optimizer による定式化	6
3.1 hxm モデリングの考え方	6
3.3 hxm モデル	7
4. hxm 言語(hxmモデルを作成する言語)	8
4.1 Function.....	8
4.2 演算子	9
4.3 言語機能	9
4.4 主なエラーメッセージ	12
【付録1】演算子一覧表	16
【付録2】LocalSolver から Hexaly への移行について	20
【付録3】Hexaly Optimizer14.5 リリースノート	25
【付録4】toy2.hxm プログラム(toy モデルに関数:output を追加).....	27
【付録5】BNF Syntax(バックス記法)	29
BNF Syntax	29

1. Hexaly Optimizer とは

Hexaly Optimizer は、実行するプログラム名としては hexaly である。最新の Hexaly Optimizer13.5 は、大規模最適化問題を実用時間内で効率よく求めることを目的とした新しいアプローチの All-In-One ソルバーであり、従来の数理計画法システムを集大成したものとなっている。とくに、スケジューリング問題、大規模組み合わせ問題に対して、大幅な性能向上を実現している。

特に、最新の Hexaly Optimizer13.5 では、大規模スケジューリング問題に対する性能強化を実現している。

Hexaly Optimizer はフランスの 6 人の若手 OR 実践者が 10 年の歳月をかけて開発したものであり、All-In-One Solver として、LP (線形計画法)、MIP (混合整数計画法)、CP(制約論理プログラミング)、NLP (非線形計画法)、LP (線形計画法) 問題を解くことができる。特に、MIP、CP では、現実的には解けなかった大規模組み合わせ最適化問題に対して、最新の各種解法を融合させて進化した解法で解くことを実現している。Hexaly_13_5 は、lsp から hxm への移行を促進するために、従来の lsp モデルファイルも実行可能である。

Hexaly Optimizer の高機能、新機能については、
<https://www.Hexaly Optimizer.com/>
で説明している。

2. Hexaly Optimizer の実行方法 (稼働確認)

**Hexaly Optimizer 13.5 は PC, Unix, MAC の 64bits モードで稼働する。
モデルファイル名は、xxx.lsp から xxx.hxm に変更する必要がある。**

Hexaly Optimizer は C++で開発されており、実行プログラムを単体で動かすだけでなく、C++、Java、C#、Python のプログラムから呼ぶことができる。

Hexaly Optimizer はインタプリター型で直接実行することができるモデリング言語 (Hexaly Modeler) (LSP) を持っている。インタプリター型で Hexaly Optimizer を実行する場合には、最新の関数型プログラミングをベースとした **hxm ファイル (xxx.hxm)** を作成し、パラメータとして hxm ファイルを指定するだけで実行可能である。

また、Hexaly Optimizer が提供するクラスライブラリを使うことで、効率よく、C++, Java, C# (.net)、python で、数理計画法システムのアプリケーションを開発することが可能である。

本ドキュメントでは、PC (64bits 版) で DOS のコマンドプロンプトから Hexaly Optimizer を直接実行する例を説明する。

Hexaly Optimizer を実行するためには、Hexaly Optimizer をインストールし、ライセンスを獲得する必要がある。その手順については、以下を参照されたい。

[https://www.msi-jp.com/Hexaly Optimizer/download/](https://www.msi-jp.com/Hexaly%20Optimizer/download/)

2.1 hxm ファイルの実行

インストールされた内容を示。

Hexaly Optimizer_13_5 フォルダ

- | - bin : 実行プログラム hexaly.exe
- | - docs : 説明書、C++, C#, Java 等のクラスライブラリの説明
- | - examples : 例題集 (hxm, C++, C#, Java, Python)
- | - include
- license.dat : ライセンスキーを設定する
- TERMS_AND_CONDITIONS
- Uninstall : アプリケーションソフト

1) 実行手順

Hexaly Optimizer を直接実行する例を説明する。

- DOS コマンドプロンプトを立ち上げる。
- bin フォルダに hxm ファイル(toy.hxm)をコピーする。
- **hxm** ファイル名を指定して、hexaly.exe を実行する。

実効例) C:\Hexaly_13_0\bin>hexaly toy. h x m lsTimeLimit=1

※hxm ファイル名 : toy.hxm。

※パラメタ lsTimeLimit は実行時間を 1 秒と指定。

2) hxm ファイルの例

ここでは、examples¥toy の **hxm** ファイル:toy.hxm を示す。
Toy モデルは、ナップサック問題である。品物が、8品あり、それぞれの重さと価値を以下に定義する。

重さ : 10、60、30、40、30、20、20、2 kg

価値 : 1、10、15、40、60、90、100、15 円

ナップサックには最大 102kg まで品物を入れることができ、価値が最大になるよう、どの品物を選べばよいか、その時の価値はいくらになるかが問題となる。

hxm による定式化を以下に示す (toy.hxm ファイルの内容)。

```
* Declare the optimization model */
function model () {

    // 0-1 decisions
    x_0 <- bool(); x_1 <- bool(); x_2 <- bool(); x_3 <- bool();
    x_4 <- bool(); x_5 <- bool(); x_6 <- bool(); x_7 <- bool();

    // Weight constraint
    knapsackWeight <- 10 * x_0 + 60 * x_1 + 30 * x_2 + 40 * x_3 + 30 * x_4 +
    20 * x_5 + 20 * x_6 + 2 * x_7;
    constraint knapsackWeight <= 102;

    // Maximize value
    knapsackValue <- 1 * x_0 + 10 * x_1 + 15 * x_2 + 40 * x_3 + 60 * x_4 +
    90 * x_5 + 100 * x_6 + 15 * x_7;
    maximize knapsackValue;
}

/* Parametrize the solver */
function param () {
    hxTimeLimit = 10;
}
```

※bool()が意志決定変数を意味し、bool()で定義された変数の0と1の値の組合せを高速に評価することで、最適解を求める。

2.2 実行結果

2.1 で示した toy モデル (toy.hxm ファイル) の実行結果を以下に示す。

```
:\hexaly_14_5>cd bin
```

```
C:\hexaly_14_5\bin>hexaly toy.hxm
```

```
Hexaly Optimizer 14.5.20260202-Win64. All rights reserved.
```

```
Load toy.hxm...
```

```
Run model...
```

```
Run param...
```

```
Run optimizer...
```

```
Model: expressions = 38, decisions = 8, constraints = 1, objectives = 1
```

```
Param: time limit = 10 sec, no iteration limit
```

```
[objective direction ]: maximize
```

```
[ 0 sec,  0 itr]:      0
```

```
[ optimality gap  ]: 100.00%
```

```
[ 0 sec, 1001 itr]:   280
```

```
[ optimality gap  ]:    0%
```

```
1001 iterations performed in 0 seconds
```

```
Optimal solution:
```

```
obj =    280
```

```
gap =    0%
```

```
bounds =   280
```

```
C:\hexaly_14_5\bin>
```

計算終了時（時間指定等で打ち切られた場合等）の最終解状態を、以下の3つで出力する。

- infeasible : 実行不可能解
- feasible : 実行可能状態
- optimal : 最適解

Hexaly Optimizer14.5 では、解空間が凸の場合には、解の上界または下界を計算し、最適解とのギャップを表示する。

添付資料5に toy2.hxm として、output function を使って答えを出力した例を示す。

3. Hexaly Optimizer による定式化

Hexaly Optimizer の定式化は意思決定変数を定義することから始まる。

意思決定変数には(bool)として定義した 0-1 変数、上下限を持つ整数変数(int)、上下限を持つ実数変数(float)、変数セット (list) として定義する変数セットの組合せからなり、意思決定変数の値を変化させ、超高速に解の探索を試行することで、大規模最適化問題を実用的に解くことが基本の考え方である。

bool 変数の数がたとえ 1000 万変数を超えても実用的な意味で解を求めることができる。bool 変数で定義した意思決定変数の組合せが解となる。解を構成する変数を一組だけ変化させ解を探索していく。ため、Hexaly Optimizer 用に定式化するためには、意思決定変数を使って、目的関数、制約関数を定義する必要がある。

ナップサック問題のように条件に合う品物を選ぶのが目的であれば、品物を選ぶか選ばないかを bool 変数として定義すれば良い。ある品物を選ばれた場合、bool 変数が 1 を取ることを想定すれば、品物の重さ及び価値を直接計算できるため、bool 変数の組合せで制約条件、目的関数を評価することができる。

Hexaly Optimizer13.5 では、従来の MIP 問題のように、上下限のある整数変数を意思決定変数として定義することができる。最適化計算途中にマテリアルバランスを求め、どこで何をいくつ作る/使うかを答えとして求めることができる。この場合、拠点配置や設備投資を決める固定費問題として、生産数量等の変数を float 変数として定義し、工場数を制限する場合生産数量を決める意思決定変数の数を LSP で制約条件として定義すればよい（従来は bigM を使用して関連付ける必要があったが技巧的な定式化は不要である）。

※ただし、既存の MIP 問題の定式化で工場制約と倉庫制約を別々の意思決定変数で定義していた場合には、解探索が必ずしも速くならないため、Hexaly Optimizer に適した問題定義が重要である。

3.1 hxm モデリングの考え方

・ 以下の手順でモデリングを行う。

- 1) 意思決定変数(bool 変数及び float 変数等)を定義する。
- 2) 上記の意思決定変数をすべて使って、制約条件、目的関数を定義する。

この時、MIP のように、線形制約にこだわる必要はなく、制約条件、目的関数ともに、非線形表現が可能である。・

3.2 hxm での bool 変数の定義

以下に典型的な問題毎に bool 変数の定義イメージを示す。

- ナップサック問題： X_p (p : 品物)
- ルート選択問題： X_r (r : ルート)
- 裁断計画問題： $X_{p,q}$ (p : 裁断パターン、 q : パターンの使用回数)
- 人員配置問題： $X_{p,t,j}$ (p : 人員、 t : 時間、 j : ジョブ)
- 車両投入計画： $X_{c,p}$ (c : 車両、 p : ポジション (順番))
- SCM： $X_{t,i,j,k,p}$ (t : 期、 i : 工場、 j : ライン、 k : 倉庫、 p : 製品)
- スケジューリング： $X_{t,i,j,p}$ (t : 時間等、 i : 工場、 j : ライン、 p : 製品等)

,

Hexaly Optimizer では数百万以上の bool 変数を定義することができ、MIP の定式化と違い、オーダ単位に意思決定変数(bool)を定義することで、より、自然な形での定式化が可能となる。

Hexaly Optimizer13.0 は、最初に事前解析で実行可能解や上界値 (下界値) を求めるため、利用者が実行不可能性または実行可能性を考慮する必要はない。

3.3 hxm モデル

hxm モデルは、以下の要素から構成される。

- 意志決定変数： `bool()`、`float` (下限値、上限値)、`int` (下限値、上限値)、`list()`
- 副生変数：任意の変数であり、プログラミングをわかりやすくすることができる。
変数の定義には、`<-` を使用する。
- 制約： `constraint` (予約語) で、制約条件を定義する。
`constraint` 制約式で定義された値が実行可能性の判定で使用される。
- 目的関数： `minimize` (予約語) または `maximize` (予約語) で目的関数を定義する。目的関数は複数定義可能であり、定義された順番に最適化を行う。目的計画法として利用可能である。

4. hxm 言語(hxmモデルを作成する言語)

hxm 言語は、従来の LSP 言語を踏襲しており、最適化問題をモデル化し、モデルの検証及び解の検証を行うフェーズでの試行錯誤を行うのに最適な環境を提供することを目的として開発されている。

hxm 言語は、最新の関数型プログラミング言語である。関数型プログラミング言語の特長は、型推論を備えた言語であるため、Java や C 言語と異なり、コンパイラが自動的にデータの種類を推定するため、データの種類 (型) をプログラマが指定する必要がない。その結果、プログラムの記述は Ruby など軽量言語のように簡潔である。

軽量言語では実現できないコンパイラによるプログラムのチェックが可能である。

hxm 言語の特徴は以下：

- 迅速に開発できる (開發生産性が良い。従来に比べて、1/5 から 1/2 の開発量)
 - バグを抑えやすい (コンパイラが型の間違い等を自動的にチェックする)
 - アプリケーションの性能を向上させやすい
 - 簡潔かつシンプルなモデリング言語 (できるだけ省略できるよう設計)
- ※大規模問題でも制約条件及びデータがそろっていれば、1日でもモデリングと実行が可能である。
- 作成(修正) ↔ 実行が同時にできる (一つはエディタ、もう一つは DOS コマンドプロンプトの二つのウインドウを操作しながら開発が可能である。
 - 目的計画法のように目的を段階的に設定することができるため、モデルの開発及び解の検証を段階的に行うことができる。

4.1 Function

hxm 言語はメインプログラムがなく、以下の5つの基本的ファンクションからなる。

`function model()`は必須であるが、その他は必要に応じて使用すればよい。

また、以下の基本的なファンクションを使用することができる。

- `input`: for declaring your data or reading them from files.
- `model`: for declaring your optimization model.
- `param`: for parameterizing the local-search solver before running.
- `display`: for displaying some info in console or in some files during the resolution.
- `output`: for writing results in console or in some files, once the resolution is finished.

4.2 演算子

hxm モデルの中で、自由に使用できる。とくに、目的関数、制約条件の記述に利用できる、非線形制約、非線形目的関数として利用可能である。演算子には、以下の種類がある。詳細は一覧表を参照されたい。

- 算術演算子 (sum、min、max、sin、cos、log、exp 等)
- 論理演算子 (not、and、or、xor)
- 関係演算子 (==、!=、<=、>=、<、>)
- 複合演算子 (if、array+at)

4.3 言語機能

1) 変数定義

変数の定義の例を示す。以下はすべて有効である。

```
a = true;    // a = 1
b = 9;       // b = 9

c = a + b;   // c = 10
c = a * b;   // c = 9

c = a == b;  // c = 0
c = a < b;   // c = 1
```

2) 配列定義

Hexaly Optimizer の配列は map で定義することができる。

Map は、値とキーを併せ持ったデータ構造になっている。キーは、整数であり、かならずしも連続的である必要はない。値は、どんなタイプでも可能であり、キーに対応させるために、[ブラケット]表記法を用いる。

```
a = map("z", 9); // a[0] = "z", a[1] = 9
a = {"z", 9};    // a[0] = "z", a[1] = 9
a["a"] = "abc"; // a[0] = "z", a[1] = 9, a["a"] = abc
```

3) 条件判定

条件判定は、if 文を使用する。記述形式は、以下：

```
if (C) S_true; else S_false;
```

または、? : で簡潔に記述することもできる。

```
if (1 < 2) c = 3; else c = 4;
c = 1 < 2 ? 3 : 4;

if (0) c = "ok";
if (true) c = "ok";
if (2) c = "error"; // ERROR: invalid condition

c = 0 * 9; // c = 0
if (c) {
    a = "L";
    b = 0;
} else { // executed block
    a = "S";
    b = 1;
}
```

4) 繰り返し

繰り返しには、while と for がある。

While は、以下で記述する。C が真である限り、S が実行される。

```
do S; while (C);
```

for は、以下で記述する。v が V にある限り、S が実行される。

```
for [v in V] S;
```

また、キーと値がセットの場合には、以下で記述する。

```
for [k, v in M] S;
```

```

for [i in 0..2] a[i] = i + 1; // a[0] = 1, a[1] = 2, a[2] = 3
s = 0; for [v in a] s = s + v; // s = 6
s = 0; for [k,v in a] s = s + k + v; // s = 9

for[i in 0..9]
  for [j in i+1..9]
    for [k in j+2..9]
      a[i][j][k] = i + j + k;
for[i in 0..9][j in i+1..9][k in j+2..9] // compact
  a[i][j][k] = i + j + k;

for[i in 0..9][j in i+1..9][k in j+2..9]
{
  a[i][j][k] = i + j + k;
  b[i][j][k] = i * j * k;
}

```

5) 繰り返し演算

繰り返し演算は、以下で記述する。

```
for [v in V] a[v] = f(v);
```

LSP では、以下の省略形で記述可能である。

```
a[v in V] = f(v);
```

```

for[i in 0..9][j in i+1..9][k in j+2..9]
  a[i][j][k] = i + j + k;

a[i in 0..9][j in i+1..9][k in j+2..9] = i + j + k;

x[i in 0..n-1][j in 0..m-1] <- bool();.

```

6) 関数

hxm では、任意に関数を定義できる。関数の値は、0 (false) または 1 (true) でも良いし、数値でも良い。hxm プログラムは通常、function 間で共通の変数定義

(global)になっているため、function内でローカルに使用したい場合には、localの宣言子でローカル変数であることを定義する必要がある。

```
function isEven(v) {
  if (v % 2 == 0) return true;
  else return false;
}

function computeSumOfEvenNumbers(a,b) {
  local total = 0;
  for [v in a..b : isEven(v)]
    total = total + v;
  return total;
}
```

4.4 主なエラーメッセージ

コマンド・ラインで絶対に必要な引数は、`hxm` ファイルの名前です。もし `hxm` ファイルが利用できないならば、エラーを出す。また、コマンド・ラインの他の全ての引数（パラメータ等）は、フォーマット `identifier=value` を持たなければならない。

- `<f>` doesn't exist or is not accessible. // `hxm` file
- Invalid argument format for `<arg>`. Expected format : `identifier=value`.

LSP 言語は型を強く意識した言語であり、関数のパラメータには正しい型が必要である。

- Function `<f>` cannot handle argument of type `<t>`. Argument of type `<t2>` is expected.
- Function `<f>` takes `<x>` argument(s) but `<y>` were provided.
- Function `<f>` : `<T>` expression expected for argument `<i>`.

同様に、型チェックで不適切な型の場合には、エラーメッセージを出力する。

- Cannot apply `<opName>` operator on type `<T>`.

- Cannot apply <opName> operator between types <T1> and <T2> .
Cannot cast <T1> to <T2>.
- Cannot apply ternary operator '?' on given operators : incorrect argument type.
- Cannot cast 'nil' to <T>. A variable or a map element may not be assigned. いくつかの関数は引数を持つ。もし、引数の数が合わない場合には、エラーを出力する。 :
- Function <f> takes at least <x> argument(s) but <y> were provided.
- Function <f> takes at most <x> argument(s) but <y> were provided.

関数を使う時、関数が未定義であれば、エラーメッセージを出力する。また、既存の関数と同じ名前の関数を定義するのもエラーである。

変数にかんしては、自由に再定義（再利用）可能である。ただし、局所変数だけは、同じ名前でもう一度使うことはできない。もし、変数が値を持たない場合は、nil の値を持つ。

- Function <f> already defined.
- Function <f> undefined.
- Variable <name> already defined.

Input/output 関数は、指定されたファイルの入出力チェックを行う。

- File <f> cannot be opened.
- Cannot read from file <f>.
- Cannot write to file <f>.

数字または文字列を入力時に、プログラムとデータが一致しない（データ数、タイプ等が一致しない）場合およびファイルの最後まで読んだ場合にはエラーを出力する。

- Cannot convert the current token to int.
- Cannot convert the current token to double.
- End of file: no more line to read from file <f>.
- End of file reached.

文字列の操作では、文字列が空でないこと及びインデックスが許容範囲内であることが必要である。

- The given index for substring is out of range. Min value: 0, Max value: <len>.
- Number of characters for substring must be greater than 0.
- Search string is empty.

マップの制限として二つある。

- キーは整数または文字であること
- イタレーション中（連続して探索計算している間）は、マップを変更してはならない:

- 'nil' provided as key for a map. The key variable may not be assigned.
- Only types 'string' and 'int' are allowed for keys in maps.
- Cannot iterate on a modified map.

hxm モデルに対してパラメータで数値を指定する場合には、許容範囲の数値でなければならぬ。

- The objective bound must be an integer, a double or a boolean for objective <objIndex>
- The objective bound must be an integer or a boolean for objective <objIndex>
- The number of threads cannot exceed 1024.
- The annealing level size must be an integer between 0 and 9.
- Advanced parameter <key> does not exist.

モデルで式または変数を表現する場合には、<- で宣言する必要である。局所変数を <- を使って宣言することは出来ない。

モデルには、必ず目的関数がなければならない。また、目的関数は数式で表現する必要がある。 マップ等は使用できない。制約式は、バイナリ表現でなければならない。

- Cannot assign Hexaly Optimizer expressions to local variables.
- At least one objective is required in the model.
- Only boolean expressions can be constrained.
- Only expressions with a value can be added in the objectives list.

setValue 関数は、意志決定変数（bool 変数）にのみ初期値を与えることができる。

- The only allowed values are 0 or 1.

$x \leftarrow a[y]$ の数式表現では、マップとしてゼロから連続した整数キーが必要となる。また、バリューとして、キーの数ぶんだけ、数値データまたは LS 表現が必要である。

- All keys must be integers. Type found: $\langle T \rangle$
- Values must be integers, booleans or expressions. Type found: $\langle T \rangle$
- The first key must be 0. Key found: $\langle \text{key} \rangle$
- Keys are not in a continuous range. Next key expected $\langle \text{key1} \rangle$. Key found: $\langle \text{key2} \rangle$.

探索の間に、変数値が制約を満足しない場合が発生することに注意されたい。例えて言えば、実行可能状態の時に、実行可能解を探索中に起きるケースがある。ゼロ割や配列オーバーフローが起きた時であり、割算の分母がゼロまたはインデックスが範囲外になったことを意味する。 $z \leftarrow x/y$ のような場合には、 $z \leftarrow x/\max(1,y)$ と表現することが望ましい。

- Division by zero.
- Index out of bounds for 'at' operator (index: $\langle \text{indexId} \rangle$, array size: $\langle n \rangle$)

以上

【付録 1】 演算子一覧表

主な演算子と関数の表

以下の表では、各演算子が hxm 言語の名前で識別されている。Python、C++、C#、または Java では、これらの名前は各言語のコーディング規則と予約済みキーワードを尊重している。

- C++ と Java では、決定には "Var" という接尾辞が付く (boolVar、floatVar、intVar、setVar と listVar)
- C# では、すべての関数が大文字で始まる

	Function	Description	Arguments type	Result type	Arity	Symb
Decisional	bool	Boolean decision variable with domain {0,1}	<i>none</i>	bool	0	
	float	Float decision variable with domain [a, b]	<i>2 doubles</i>	double	2	
	int	Integer decision variable with domain [a, b]	<i>2 integers</i>	int	2	
	interval	Interval decision variable with domain [minStart, maxEnd)	<i>2 integers</i>	interval	2	
	list	Ordered collection of integers within a range [0, n - 1]	<i>1 integer</i>	collection	1	
	set	Unordered collection of integers within a range [0, n - 1]	<i>1 integer</i>	collection	1	
Arithmetic	sum	Sum of all operands	bool, int, double	int, double	n >= 0	+
	sub	Substraction of the first operand by the second one	bool, int, double	int, double	2	-
	prod	Product of all operands	bool, int, double	int, double	n >= 0	*
	min	Minimum of all operands	bool, int, double	int, double	n > 0	
	max	Maximum of all operands	bool, int, double	int, double	n > 0	
	div	Division of the first operand by the second one	bool, int, double	double	2	/
	mod	Modulo: $\text{mod}(a, b) = r$ such that $a = q * b + r$ with q, r integers and $r < b$.	bool, int	int	2	%
	abs	Absolute value: $\text{abs}(e) = e$ if $e \geq 0$, and $-e$ otherwise	bool, int, double	int, double	1	

	Function	Description	Arguments type	Result type	Arity Symb
	dist	Distance: $\text{dist}(a, b) = \text{abs}(a - b)$	bool, int, double	int, double	2
	sqrt	Square root	bool, int, double	double	1
	cos	Cosine	bool, int, double	double	1
	sin	Sine	bool, int, double	double	1
	tan	Tangent	bool, int, double	double	1
	log	Natural logarithm	bool, int, double	double	1
	exp	Exponential function	bool, int, double	double	1
	pow	Power: $\text{pow}(a, b)$ is equal to the value of a raised to the power of b .	bool, int, double	double	2
	ceil	Ceil: round to the smallest following integer	bool, int, double	int	1
	floor	Floor: round to the largest previous integer	bool, int, double	int	1
	round	Round to the nearest integer: $\text{round}(x) = \text{floor}(x + 0.5)$.	bool, int, double	int	1
	scalar	Scalar product between 2 arrays.	array	int, double	2
	piecewise	Piecewise linear function product between 2 arrays.	array, int, double	double	3
Logical	not	Not: $\text{not}(e) = 1 - e$.	bool	bool	1 !
	and	And: equal to 1 if all operands are 1, and 0 otherwise. Takes value 1 when applied to an empty collection.	bool	bool	$n \geq 0$ &&
	or	Or: equal to 0 if all operands are 0, and 1 otherwise. Takes value 0 when applied to an empty collection.	bool	bool	$n \geq 0$
	xor	Exclusive or: equal to 0 if the number of operands with value 1 is even, and 1 otherwise. Takes value 0 when applied to an empty collection.	bool	bool	$n \geq 0$
Relational	eq	Equal to: $\text{eq}(a, b) = 1$ if $a = b$, and 0 otherwise	bool, int, double	bool	2 ==
	neq	Not equal to: $\text{neq}(a, b) = 1$ if $a \neq b$, and 0 otherwise	bool, int, double	bool	2 !=

Function	Description	Arguments type	Result type	Arity	Symb
geq	Greater than or equal to: $\text{geq}(a, b) = 1$ if $a \geq b$, 0 otherwise	bool, int, double	bool	2	\geq
leq	Lower than or equal to $\text{leq}(a, b) = 1$ if $a \leq b$, 0 otherwise	bool, int, double	bool	2	\leq
gt	Strictly greater than: $\text{gt}(a, b) = 1$ if $a > b$, and 0 otherwise. In case of intervals: $\text{gt}(a, b) = 1$ if $\text{start}(a) > \text{end}(b)$, and 0 otherwise.	bool, int, double, interval	bool	2	$>$
lt	Strictly lower than: $\text{lt}(a, b) = 1$ if $a < b$, and 0 otherwise. In case of intervals: $\text{lt}(a, b) = 1$ if $\text{end}(a) < \text{start}(b)$, and 0 otherwise.	bool, int, double, interval	bool	2	$<$
Conditional iif	Ternary operator: $\text{iif}(a, b, c) = b$ if a is equal to 1, and c otherwise	bool, int, double	bool, int, double	3	$?:$
Set related					
count	Returns the number of elements in a collection.	collection, interval, array	int	1	
indexOf	Returns the index of a value in a collection or -1 if the value is not present.	collection, int	int	2	
contains	Returns 1 if the collection contains the given value or 0 otherwise.	collection or interval, int	bool	2	
partition	Returns true if all the operands form a partition of their common domain.	collection	bool	$n > 0$	
disjoint	Returns true if all the operands are pairwise disjoint.	collection	bool	$n > 0$	
cover	Returns true if all the operands form a cover of their common domain.	collection	bool	$n > 0$	
array	Creates an array of fixed or variadic size.	bool, int, double, array, list, set	array	$n \geq 0$	
stepArray	Creates an stepArray of fixed size.	bool, int, double	array	$n \geq 1$	
at	Returns the value in an array or a list at a specified position.	array, list, int	bool, int, double	$n \geq 2$	$[]$
find	Returns the position of the first collection containing the given element in the array, or -1 if the value is not present.	array, int	int	2	
sort	Returns the array sorted in ascending order. When used with two arguments, the array is	array, lambda	array	1 or 2	

	Function	Description	Arguments type	Result type	Arity Symb
		sorted based on the values returned by the lambda.			
	distinct	Returns the unordered set of distinct values in an array. When used with two arguments, the distinct values are based on the values returned by the lambda applied to the iterable.	array, list, set, interval, lambda	set	1 or 2
	intersection	Returns the unordered set of values present in both iterables.	array, list, set	set	2
Interval related	start	Returns the start of a non-void interval.	interval	int	1
	end	Returns the end of a non-void interval.	interval	int	1
	length	Returns the length of a non-void interval, equivalent to $\text{end}(\text{interval}) - \text{start}(\text{interval})$.	interval	int	1
	hull	Returns the smallest interval including all the intervals given in operands.	interval	interval	$n \geq 0$
Other	call	Call a function. It can be used to implement your own operator.	bool, int, double	double	$n > 0$

【付録2】LocalSolver から Hexaly への移行について

2023年11月より、LocalSolverはHexalyとなりました。この新しい名称は、私たちのソルバーが、局所探索技術に基づくソルバーから、多くの最適化技術を組み合わせた全体最適化ソルバーへと徐々に移行してきたことを、よりよく反映することを意図しています。この新しい名称の詳細については、ブログの該当ページをご覧ください。

商用コンテンツに加え、すべてのAPI、バイナリ、インストーラを変更することで、このアイデンティティを既存製品の中心に展開することを決定しました。私たちは、私たちと共にこのステップを踏み、それに従ってソフトウェアを適応させるよう、すべてのユーザーを招待します。いくつかのクラス名と関数名に限定されますが、私たちはこの変更がお客様にとって大きな変化をもたらす可能性があることを認識しています。そのため、円滑な移行を確実にするため、旧APIは1年間の移行期間中サポートされます。この移行に必要なすべての情報と、対応するスケジュールは以下の通りです。

スケジュール

LocalSolver から Hexaly への移行は3段階に分けて行われます：

- Hexaly 12.5 (2023年11月)： ウェブサイトと製品名の変更。旧LocalSolverの商標は消え、Hexalyに変わるが、技術的には何も変わらない（同じインストーラ、同じバイナリ、同じAPI）。
- Hexaly 13.0 (2024年7月)： 私たちのAPIとバイナリは新しいアイデンティティに移行します。スムーズな移行を保証するために、古いAPIはまだサポートされていますが、新しい機能は追加されません。古いAPIは、あなたのコードを変更することなく、最新のパフォーマンス強化やバグ修正の恩恵を受けることができます。
- Hexaly 14.0 (mid-2025)： 古いLocalSolver APIは完全に消滅します。Hexalyへの移行が完了します。

備考

LocalSolverでデプロイされたアプリケーションはどうなりますか？

旧バージョンのLocalSolverでローカルまたはクラウドにデプロイされたアプリケーションは引き続き動作し、新しいライセンスはこれらのアプリケーションに有効です。ただし、アプリケーションでLocalSolverのバグに遭遇した場合（稀なケースですが）、最新のHexalyバージョンにアップグレードする必要があります。以下に説明するように、この移行は簡単で、数分で行うことができます。

インストーラとパッケージ

インストーラは「Hexaly」という新しい名前に変更され Hexaly ブランドのインストーラは 1 つだけですが、C#、Java、Python の両方のバージョンの API がインストールされます (Hexaly 14.0 以降のバージョンではそうではありません)。

Hexaly Optimizer のインストールを自動化している場合は、デプロイスクリプトを少し変更する必要があります。「LocalSolver」を「Hexaly」に変更してスクリプトを更新する必要があります。

環境変数

LocalSolver とそのライセンスの場所は、3 つの環境変数 LS_HOME (Windows のみ)、LS_LICENSE_PATH、LS_LICENSE_CONTENT によって制御されていました。これらの変数は現在、それぞれ HX_HOME、HX_LICENSE_PATH、HX_LICENSE_CONTENT になっています。古い変数も引き続き動作しますが、両方が指定されている場合は、新しい変数よりも優先順位が低くなります。

Pip パッケージ (ホイール)

移行フェーズでは、2 つの Python ホイール (pip install 可能なパッケージ) を配布します。pip install localsolver や pip install hexaly を使用することはできますが、古い Python API には新しい機能が追加されないことに留意してください。しかし、あなたのコードや既存の Python プロジェクトの行を変更することなく、オプティマイザの最新のパフォーマンス拡張やバグ修正の恩恵を受けることができます。

API の変更

私たちのリブランディングは、LocalSolver から Hexaly への API の完全な名称変更を伴います。これらの新しい API によってもたらされる変更は、以下に詳述されています。要するに、私たちは主に 2 つの主要なクラスの名前と接頭辞 LS を Hx に変更しました。

主な変更点は以下の通りです：

- メインクラス名 LocalSolver は HexalyOptimizer になります。
- メインクラス名 LSPModeler は HexalyModeler となります。
- 全てのクラスの LS 接頭辞が Hx に変更された (小文字の Hx に注意)。
- すべてのクラスの LSP 接頭辞が Hxm に変更されました (大文字と小文字の Hxm に注意してください)。

モデラーとオプティマイザの名前空間が 2 つになりました。以前は、localsolver 名前空間とその子 localsolver.modeler 名前空間がありました。こ

の2つは分離され、親ではなく、`hexaly.optimizer` と `hexaly.modeler` という兄弟になりました。

長い間非推奨だった機能やメソッドがいくつか削除されました。

一般的に、`ls` と `lsp` への参照はすべてなくなりました。

言語固有の変更も行われました。詳細は以下の通りです。

新しい API にすぐに切り替えたくない場合は、古い Python、Java、C# API を使うこともできます。しかし、これらの古い API には新しい機能は追加されないことを覚えておいてください。とはいえ、これらの API を使用することで、コードの1行も変更することなく、パフォーマンスの向上やバグ修正の恩恵を受けることができます。

C# API の変更

バイナリ (DLL) の名前が `localsolvernet.dll` から `Hexaly.NET.dll` に変更されました (大文字と小文字は重要です)。この変更は、C# ライブラリの命名規則をよりよく反映します。

`localsolver` 名前空間は `Hexaly.Optimizer` になりました。

名前空間 `localsolver.modeler` は `Hexaly.Modeler` になります。

これまで `GetLocalSolver()` または `CreateLocalSolver()` と呼ばれていたパブリック・メソッドは、`GetOptimizer()` および `CreateOptimizer()` という名前に変更されました。

Java API の変更

`localsolver.jar` が `hexaly.jar` になりました。

`localsolver` パッケージが `com.hexaly.optimizer` になりました。

`localsolver.modeler` パッケージは `com.hexaly.modeler` になります。

新しい名前は、Java の慣例 (会社のインターネット・ドメイン名の逆を使用) に沿っています。

以前は `getLocalSolver()` または `createLocalSolver()` と呼ばれていたパブリック・メソッドは、`getOptimizer()` および `createOptimizer()` という名前に変更されました。

Python API の変更

`localsolver` モジュールが `hexaly.optimizer` になりました。

`localsolver.modeler` モジュールが `hexaly.modeler` になりました。

`localsolver` (旧 API) 用と `hexaly` 用の2つの pip パッケージ (wheel) を維持します。

以前は `get_localsolver()` や `create_localsolver()` と呼ばれていたパブリック・メソッドは、`get_optimizer()` や `create_optimizer()` という名前に変更されました。

C++ API の変更

他の言語とは異なり、従来の 2 つの名前空間ではなく、1 つの名前空間を選択しました。そのため、`localsolver` と `localsolver::modeler` は単に `hexaly` となります。新しい C++ API だけが提供されることに注意してほしい。`localsolver` という名前で使われていた古い API は、永久に廃止されました。

バイナリ / 実行可能ファイル

API と異なり、実行ファイルとネイティブ・ライブラリは、`hexaly / localsolver` という二重のネーミングでは提供されません。次のリリースでも古い `LocalSolver` API を維持したい場合でも、プロジェクトの依存関係として統合するバイナリを変更する必要があります。これには、デプロイスクリプトのちょっとした変更が必要になるかもしれません。

`localsolver.exe` は `hexaly.exe` になります。

`lskeygen.exe` は `Hexaly 13.0` では `hxkeygen.exe` となり、`Hexaly 13.5` では永久に削除されます（このバージョンから、ライセンスチェックはハードウェア署名に基づくものではなくなりました）。

`lstokenserver.exe` は `Hexaly 13.0` で `hxtokenserver.exe` となり、`Hexaly 13.5` で永久に削除されます（フローティング ライセンスは数年前から非推奨となり、公式には配布されていません）。

`localsolver130.dll / liblocalsolver130.so / liblocalsolver130.dylib` は `hexaly130.dll / libhexaly130.so / libhexaly130.dylib` になります。

モデリング言語 (hxm)

これまで `.lsp` という拡張子だったファイルは、`.hxm` (`HeXaly Modeler`) という拡張子に変更されました。`Hexaly` は引き続き `.lsp` ファイルを問題なく読み込むことができます。

`lsTimeLimit`、`lsNbThreads`、`lsSeed...` のようなグローバル変数の名前は、以前は `ls` が前についていましたが、現在は `hx` が前についています

(`hxTimeLimit`、`hxNbThreads`、`hxSeed`)。古い名前もサポートされていますが、新しい名前が優先されます。言い換えると、`hxSeed` と `lsSeed` の両方を入力した場合、`hx` が接頭辞の変数だけが考慮されます。

以前は `localsolver` と呼ばれていたモジュールは、現在は `hexaly` と呼ばれています。

古い localsolver モジュールはまだ存在しますが、新しい hexaly モジュールのエイリアスに過ぎません。

要するに、すべての LSP ファイルはコードを変更することなく機能し続けます。呼び出される実行ファイルの名前だけが localsolver.exe から hexaly.exe に変わります。Java、C#、Python のデュアル API が 1 年で廃止されるのとは異なり、これらの古い .lsp ファイルや古いグローバル変数のサポートを削除する予定はありません。これらは単に時間の経過とともに非推奨となるだけです。

LSB ファイル

ファイル拡張子が .lsb から .hxb に変わる以外は何も変わりません。Hexaly は引き続き .lsb フォーマットを読み込みますし、将来のバージョンでそのサポートを削除する予定はありません。一言で言えば、次の 2 つのバージョン 13.0 と 13.5 (1 年間の移行) では、Java、C#、Python 用のデュアル API (Hexaly と LocalSolver) をユーザーに提供することにしました。この移行期間中も、Hexaly と LocalSolver の両方の API を使用することができます。

以上

【付録3】 Hexaly Optimizer14.5 リリースノート

[ジュリアン](#) 科学部長 2026年2月2日

ルーティング、スケジューリング、パッキングなど多くの問題でパフォーマンスが向上し続けている **Hexaly Optimizer 14.5** のリリースを発表できることを嬉しく思います([ベンチマーク](#)参照)。

パフォーマンス向上に関しては、以下に 2 種類のギャップが挙げられます。単に「ギャップ」と表記すると、それは最先端(SOTA)とのギャップ、すなわち Hexaly が標準サーバー(AMD Ryzen 7 7700 プロセッサ、8 コア、3.8GHz、32MB キャッシュ、32GB RAM)で計算した解と、より高性能なハードウェア上で専用アルゴリズムで数日以内に計算された最良の解との間の相対的な%の差を意味します。「最適性ギャップ」と呼ばれるとき、最適性へのギャップ、すなわち、最良の実現可能な解とソルバーが見つけた最良の双対境界との相対的なパーセントギャップを指します。

ルーティング

Hexaly 14.5 は、さまざまな制約や目的を持つ**車両ルーティング問題(VRP)**に対して大幅な性能向上をもたらします。

- [タイムウィンドウ付き賞品収集車両ルート\(PCVRPTW\)](#):1,000 人の顧客を対象とした内部ベンチマークで 1 分間の平均差 **2.7%**。
- [複数回容量車両ルーティング問題\(MTCVRP\)](#):最大 200 人の顧客を対象とした内部ベンチマークで、1 分間の平均ギャップが **2.4%**。

バージョン 14.5 は、複雑なタイミング制約や目的(例:ルート振幅)を持つ**実際の車両ルーティングアプリケーション**において、大幅な性能向上を提供します。

各種車両ルーティング問題の下限値は、**カラム生成安定化技術**のおかげでベンチマークインスタンスの 29%で改善されました。Hexaly は、最大 200 の顧客に対して、容量、異種機車、時間窓など様々な制約がある場合に対して、**平均 7.1%の最適性ギャップ**を持つソリューションを **10 分**で返品します。Hexaly は、前バージョンと比較して **13 件**の新しいベンチマークインスタンスに対して **10 分**で最適であることを証明しました。

スケジューリング

Hexaly 14.5 はマルチモードスケジューリング、累積スケジューリング、賞品収集のジョブショップ問題のバリエーションに対していくつかの改良が施されています。

- [マルチモードリソース制約プロジェクトスケジューリング\(MRCPSP\)](#):MMLIB50 ベンチマークで 1 分間の平均差が **0.9%**、MMLIB100 ベンチマークで 1 分あたり **1.7%**、最大 100 タスクの MMLIB+ベンチマークで 1 分間に **3.5%**の平均ギャップがあります。

Hexaly 14.5 は、**カレンダー制約**を伴う**実際の累積スケジューリング問題**に対して大幅な性能向上を提供します:

constraint and(0...makespan,

```
t => sum[i in 0...nbTasks](weight[i] * calendar[i][t] * contains(tasks[i], t)) <= capacity);
```

Hexaly 14.5 はまた、**実際の選言スケジューリング問題**に対しても大幅な性能向上を提供します。

梱包

Hexaly 14.5 は、**交差**演算子を用いてモデリングされた**衝突を伴うビンパッキング問題(BPPC)**のパフォーマンス向上を導入します。これは最大 1,000 品目で**マルチバインスタン**スで 1 分間の平均 **0.3%**のギャップを検出します。

非線形

Hexaly の新しい **Interior Point** アルゴリズムは、前バージョンで導入され、より高速かつ堅牢になっています。以下の結果は、ラン開始から 1 分以内に得られます:

- [ポートフォリオ最適化](#): 最大 4,000 変数を含む 270 回の内部ベンチマークのうち 210 回で最適解を見つけました。
- **CUTEst** ベンチマーク: 1,059 件中 983 件で 1%未満のギャップを持つ実現可能な解決策。

モデラー&API

Hexaly 14.5 はハル演算子を拡張し、変動構造を受け入れるようにし、ユーザーが特定のリソースに対してスケジュールされた間隔のハルを取ることを可能にします。

```
function model() {  
  l <- list(NB_TASKS);  
  x[0...NB_TASKS] <- interval(0, 100);  
  h <- hull(l, i => x[i]); // hull of the intervals selected in the list l  
  minimize length(h);  
}
```

テンプレート

ルーティング、スケジューリング、パッキングなどの問題に対応する新しいコードテンプレートが利用可能で、Hexaly をスムーズに使い始めるのに役立ちます:

[州との柔軟なリソース制約プロジェクトスケジューリング問題\(FRCSPSPS\)](#)

[広告キャンペーンの選択問題](#)

[リソース可用性コスト問題\(RACP\)](#)

[並列スケジューリング](#)

ルーティング、スケジューリング、パッキング、その他多くの問題カテゴリで性能が向上し続けている Hexaly 14.0 のリリースを喜んでお知らせします(ベンチマーク参照)。さらに、Hexaly 14.0 はいくつかの新しいモデリング機能を提供しています。

パフォーマンス向上に関しては、以下に 2 種類のギャップが挙げられます。単に「ギャップ」と表記すると、それは最先端(SOTA)とのギャップ、すなわち Hexaly が標準サーバー(AMD Ryzen 7 7700 プロセッサ、8 コア、3.8GHz、32MB キャッシュ、32GB RAM)で計算した解と、より高性能なハードウェア上で専用アルゴリズムで数日以内に計算された最良の解との間の相対的な%の差を意味します。「最適性ギャップ」と呼ばれるとき、最適性へのギャップ、すなわち、最良の実現可能な解とソルバーが見つけた最良の双対境界との相対的なパーセントギャップを指します。

【付録4】 toy2.hxm プログラム (toy モデルに関数: output を追加)

1. toy2.hxm プログラム

Toy2.hxm プログラムは、**toy2.lsp** ファイルを **toy2.hxm** に変更しただけで、中身は、全く同じである。

```
/****** toy2.hxm *****/ function model()
{
  nbProducts = 8;
  value = {1,10,15,40,60,90,100,15};
// 0-1 decisions
  x[i in 0..nbProducts-1] <- bool();
  // weight constraint
  knapsackWeight <- 10*x[0] + 60*x[1] + 30*x[2] + 40*x[3] + 30*x[4] + 20*x[5] +
20*x[6] + 2*x[7];
  constraint knapsackWeight <= 102;
  // maximize value
  knapsackValue <- 1*x[0] + 10*x[1] + 15*x[2] + 40*x[3] + 60*x[4] + 90*x[5] +
100*x[6] + 15*x[7]; maximize knapsackValue;
}

function output()
{
  println("Selected Products:");
  for [i in 0..nbProducts-1 : x[i].value == 1] println("#"+i+" (" +value[i]+")");
}
```

2. toy2.hxm 実行結果

```
C:\hexaly_14_5\bin>
C:\hexaly_14_5\bin>hexaly toy2.hxm
Hexaly Optimizer 14.5.20260202-Win64. All rights reserved.
Load toy2.hxm...
Run model...
Run optimizer...
```

```
Model: expressions = 38, decisions = 8, constraints = 1, objectives = 1
Param: no time limit, no iteration limit
```

```
[objective direction ]: maximize
```

```
[ 0 sec,  0 itr]:      0
[ optimality gap  ]: 100.00%
[ 0 sec, 1046 itr]:   280
[ optimality gap  ]:    0%
```

```
1046 iterations performed in 0 seconds
```

```
Optimal solution:
obj  =    280
gap  =     0%
bounds =    280
```

```
Run output...
```

```
Selected Products:
```

```
#2 (15)
#4 (60)
#5 (90)
#6 (100)
#7 (15)
```

```
C:\hexaly_14_5\bin>
```

【付録5】BNF Syntax(バックス記法)

BNF の表記は次のような導出規則の集合である。

```
<symbol> ::= <expression with symbols>
```

左辺の<symbol>は単一の記号である。また、<expression with symbols> は記号列、または選択を表す**バーティカルバー**「|」で区切られた記号列であり、左辺の `<symbol>` の置換となるものを表している。なお、導出規則で使用された記号は「**非終端記号**」と「**終端記号**」に分類される。導出規則群の左辺に現れた記号は「**非終端記号**」と呼ばれ、いずれの導出規則の左辺にも現れなかった記号は「**終端記号**」と呼ばれる。

BNF Syntax

This is the full grammar of the Hexaly modeling language.

BNF Syntax

identifier

```
: simple_identifier  
| contextual_keyword
```

contextual_keyword

```
: 'pragma'  
| 'as'  
| 'from'  
| 'extends'  
;
```

expression

```
: ternary_expression  
| lambda_expression  
| table_expression  
;
```

expression_no_range

```
: ternary_expression_no_range  
| lambda_expression
```

```

| table_expression
;

ternary_expression_no_range
: or_expression_no_range
| or_expression_no_range
  '?'ternary_expression_no_range
  ':'ternary_expression_no_range
;

or_expression_no_range
: and_expression_no_range
| or_expression_no_range '||' and_expression_no_range
;

and_expression_no_range
: equality_expression_no_range
| and_expression_no_range '&&' equality_expression_no_range
;

equality_expression_no_range
: relational_expression_no_range
| equality_expression_no_range '==' relational_expression_no_range
| equality_expression_no_range '!=' relational_expression_no_range
;

relational_expression_no_range
: additive_expression
| relational_expression_no_range 'is' additive_expression
| relational_expression_no_range 'is' 'nil'
| relational_expression_no_range 'is' 'bool'
| relational_expression_no_range 'is' 'int'
| relational_expression_no_range 'is' 'double'
| relational_expression_no_range '<' additive_expression
| relational_expression_no_range '>' additive_expression
| relational_expression_no_range '<=' additive_expression
| relational_expression_no_range '>=' additive_expression
;

ternary_expression
: or_expression
| or_expression
  '?' ternary_expression
  ':' ternary_expression

```

```

;

or_expression
: and_expression
| or_expression '||' and_expression
;

and_expression
: equality_expression
| and_expression '&&' equality_expression
;

equality_expression
: relational_expression
| equality_expression '==' relational_expression
| equality_expression '!=' relational_expression
;

relational_expression
: range_expression
| relational_expression 'is' range_expression
| relational_expression 'is' 'nil'
| relational_expression 'is' 'bool'
| relational_expression 'is' 'int'
| relational_expression 'is' 'double'
| relational_expression '<' range_expression
| relational_expression '>' range_expression
| relational_expression '<=' range_expression
| relational_expression '>=' range_expression
;

range_expression
: additive_expression
| additive_expression '!' additive_expression
| additive_expression '...' additive_expression
;

additive_expression
: multiplicative_expression
| additive_expression '+' multiplicative_expression
| additive_expression '-' multiplicative_expression
;

multiplicative_expression

```

```
: unary_expression  
| multiplicative_expression '*' unary_expression  
| multiplicative_expression '/' unary_expression  
| multiplicative_expression '%' unary_expression  
;
```

```
unary_expression  
: function_call_expression  
| new_expression  
| '+' unary_expression  
| '-' unary_expression  
| '!' unary_expression  
| 'typeof' unary_expression  
;
```

```
new_expression  
: 'new' member_array_expression arguments  
;
```

```
function_call_expression  
: primary_expression  
| super_expression  
| function_call_expression '[' expression ']'  
| function_call_expression '.' identifier  
| function_call  
;
```

```
member_array_expression  
: primary_expression  
| super_expression  
| member_array_expression '[' expression ']'  
| member_array_expression '.' identifier  
;
```

```
super_expression  
: 'super' '.' identifier  
;
```

```
primary_expression  
: assignment_identifier  
| 'true'  
| 'false'  
| 'nan'  
| 'inf'
```

```
| 'nil'  
| string  
| integer  
| double  
| '(' expression ')'  
;
```

```
lambda_expression  
: identifier '=>' block_statement  
| function_arguments_declaration '=>' block_statement  
| identifier '=>' lambda_body_expression  
| function_arguments_declaration '=>' lambda_body_expression  
| 'function' function_arguments_declaration block_statement  
;
```

```
lambda_body_expression  
: ternary_expression  
| lambda_expression  
;
```

```
table_expression  
: '{ '}'  
| '{ table_list }'  
;
```

```
table_list  
: expression  
| table_key '=' expression  
| table_key ':' expression  
| table_list ',' expression  
| table_list ',' table_key '=' expression  
| table_list ',' table_key ':' expression  
;
```

```
table_key  
: string  
| identifier  
| integer  
| '-' integer  
;
```

```
function_call  
: function_call_expression arguments  
| function_call_expression variadic_arguments
```

```

;

arguments
: '(' ')'
| '(' function_argument_list ')'
;

variadic_arguments
: variadic_compositor_list '(' function_variadic_list ')'
;

function_argument_list
: expression
| function_argument_list ',' expression
;

function_variadic_list
: expression
| function_variadic_list ',' expression
;

variadic_compositor_list
: '[' filter_iterator ']'
| variadic_compositor_list '[' filter_iterator ']'
;

filter_iterator
: identifier 'in' expression ':' expression
| identifier ',' identifier 'in' expression ':' expression
| identifier 'in' expression
| identifier ',' identifier 'in' expression
;

range_iterator
: additive_expression '..' additive_expression
| additive_expression '...' additive_expression
;

statement
: block_statement
| assignment_statement
| local_assignment_statement
| local_statement
| if_else_statement

```

```
| for_statement  
| while_statement  
| dowhile_statement  
| continue_statement  
| break_statement  
| modifier_statement  
| throw_statement  
| trycatch_statement  
| with_statement  
| function_call_statement  
| new_statement  
| return_statement  
| super_constructor_statement  
| ';' ;
```

```
block_statement  
: '{ '  
| '{ statement_list }'  
;
```

```
statement_list  
: statement  
| statement_list statement  
;
```

```
assignment_statement  
: identifier assignment_operator expression ';' ;  
| assignment_identifier assignment_compositor_list assignment_operator expression  
' ;'  
;
```

```
assignment_identifier  
: identifier  
| 'this'  
;
```

```
assignment_operator  
: '='  
| '<-'  
| '+='  
| '-='  
| '/='  
| '*='
```

```

    | '%='
    ;

assignment_compositor_list
: assignment_compositor
| assignment_compositor_list assignment_compositor
;

assignment_compositor
: '[' filter_iterator ']'
| '[' range_iterator ']'
| '[' expression_no_range ']'
| '!' identifier
;

local_assignment_statement
: 'local' identifier local_assignment_operator expression ';'
| 'local' assignment_identifier assignment_compositor_list local_assignment_operator
expression ';
;

local_assignment_operator
: '='
| '<-'
;

local_statement
: 'local' identifier ';'
;

if_else_statement
: if_condition statement
| if_condition statement statement
;

if_condition
: 'if' '(' expression ')'
;

for_statement
: 'for' for_compositor_list statement
;

for_compositor_list

```

```

: '[' filter_iterator ']'
| for_compositor_list '[' filter_iterator ']'
;

while_statement
: 'while' '(' expression ')' statement
;

dowhile_statement
: 'do' statement 'while' '(' expression ')' ';'
;

continue_statement
: 'continue' ';'
;

break_statement
: 'break' ';'
;

modifier_statement
: modifier expression ';'
;

modifier
: 'minimize'
| 'maximize'
| 'constraint'
;

throw_statement
: 'throw' expression ';'
| 'throw' ';'
;

trycatch_statement
: 'try' statement 'catch' '(' identifier ')' statement
;

with_statement
: 'with' '(' with_resource ')' statement
;

with_resource

```

```

: identifier
| identifier '=' expression

function_call_statement
: function_call ';'
;

new_statement
: new_expression ';'

return_statement
: 'return' ';'
| 'return' expression ';'
;

super_constructor_statement
: 'super' arguments ';'
;

declaration_list
: function_declaration
| class_declaration
| declaration_list function_declaration
| declaration_list class_declaration
;

function_declaration
: 'function' identifier function_arguments_declaration block_statement
;

function_arguments_declaration
: '(' ')'
| '(' function_identifier_list ')'
;

function_identifier_list
: identifier
| function_identifier_list ',' identifier
;

class_declaration
: class_header '{' '}'
| class_header '{' class_member_list '}'
;

```

```

class_header
: 'class' identifier
| 'class' identifier 'extends compound_name'
| 'final' 'class' identifier
| 'final' 'class' identifier 'extends compound_name'
;

class_member_list
: class_member_list class_member
| class_member
;

class_member
: class_constructor
| class_method
| class_field
| class_static_function
| class_static_field
;

class_constructor
: 'constructor' function_arguments_declaration block_statement
;

class_method
: 'override' identifier function_arguments_declaration block_statement
| identifier function_arguments_declaration block_statement
;

class_field
: identifier ';'
| identifier '=' expression ';'
;

class_static_function
: 'static' identifier function_arguments_declaration block_statement
;

class_static_field
: 'static' identifier ';'
| 'static' identifier '=' expression ';'
;

```

```
pragma_list
  : pragma_statement
  | pragma_list pragma_statement
  ;
```

```
pragma_statement
  : 'pragma' simple_identifier ';'
  | 'pragma' simple_identifier simple_identifier ';'
  | 'pragma' simple_identifier integer ';'
  | 'pragma' simple_identifier double ';'
  ;
```

```
use_list
  : use_statement
  | use_list use_statement
  ;
```

```
use_statement
  : TOKEN_USE simple_identifier ';'
  | TOKEN_USE compound_name 'as' simple_identifier ';'
  | TOKEN_USE import_list 'from' compound_name ';'
  ;
```

```
import_list
  : simple_identifier
  | simple_identifier 'as' simple_identifier
  | import_list ',' simple_identifier
  | import_list ',' simple_identifier 'as' simple_identifier
  ;
```

```
compound_name
  : compound_name_part
```

```
compound_name_part
  : simple_identifier
  | compound_name_part '.' simple_identifier
  ;
```

```
header_section
  : pragma_list
  | pragma_list use_list
  | use_list
  ;
```

```
program
: <EOF>
| declaration_list <EOF>
| header_section <EOF>
| header_section declaration_list <EOF>
;
```

以上